LOCAL COHOMOLOGY AND SHEAF COHOMOLOGY

Questions

(1) Let G be a graph, and let $I \subset S$ be the corresponding edge ideal (i.e. generated by quadric monomials $x_i x_j$, where (i, j) is an edge of the graph). Feel free to also try other Stanley-Reisner ideals too.

  (a) Compute, using local duality and approximation, the local cohomology modules of S/I (and/or their graded k-duals), for specific choices of graphs G.

  (b) These modules are $\mathbb{Z}^n$-graded (n the number of variables). Find their multi-graded hilbert series.

  (c) Consider the Cech complex (tensored with S/I). Consider the d-th graded piece (d a multidegree). Is this complex over k recognizable as the complex corresponding to some simplicial complex? Can you find a formula for the dimensions of the multi-graded parts of the local cohomology modules?

Hint. Use the `DegreesRank` or `Degrees` options when creating a polynomial ring, if you wish to compute multigraded Hilbert series.

(2) Continuing with edge ideals, we investigate the cohomology of the normal sheaf $N = Hom_S(I, S/I)$.

  (a) For your examples, find $H^0(\widetilde{N})$, and $H^1(\widetilde{N})$, and the other H^i too. Can you find a formula in terms of the properties of G?

  (b) Is $N = H^0_*(\widetilde{N})$? If not, find it.

(3) Write a Macaulay2 function which takes an S-module M as input and returns the list $\{\dim H^0(\widetilde{M}), \ldots, \dim H^n(\widetilde{M})\}$. Try your function on the following modules, (where e.g. $(n, d, p) = (3, 4, 1), (3, 4, 2), (3, 5, 1), \ldots$). Can you find H^0_* of these modules? What are their resolutions.

```
om = (n,d,p) -> (
  S = ZZ/32003[x_0..x_n];
  I = ideal random(S^1, S^{-d});
  M = omega I;
  exteriorPower(p,M))
```

Date: 25 July 2006.

(4) Write a function to compute, for a variety of dimension 2 (or 3 if you prefer, or even in general) the Hodge diamond. Apply it to ideals generated by random degree d forms in 4 or 5 variables, or to Fermat hypersurfaces (sum of the d-th powers of all of the variables).

(5) Let $X = V(f)$ be a hypersurface. Try to find in terms of f a presentation of the module M which defines Ω^1_X.

(6) [**Mystery variety**] Castelnuovo proved a great theorem: Let X be a smooth projective surface. Then X is rational if and only if $H^0(\omega_X \otimes \omega_X) = H^1(\mathcal{O}_X) = 0$, where ω_X is the sheaf corresponding to the module $Ext^{codim X}(S/I, S(-n-1))$.
  (a) Use this to investigate whether the ideal in the file "mystery.m2" defines a rational surface (you may assume that the ideal defines a smooth projective surface).
  (b) Extra credit: can you identify this surface, and/or compute its Hodge diamond?

Macaulay 2 examples from the morning lecture

```
--------------------------------------
-- Example: rational quartic curve --
--------------------------------------
S = QQ[a..d]
P1 = QQ[s,t]
F = map(P1,S,{s^4,s^3*t,s*t^3,t^4})
I = ker F
C = res (S^1/I)
C = res I
C.dd
codim I
degree I
genus I

Ext^1(S^1/I,S)
Ext^2(S^1/I,S)
Ext^3(S^1/I,S)

Ext^2(S^1/I,S^{-4})
Ext^3(S^1/I,S^{-4})

--------------------------------------
-- Local cohomology -----------------
--------------------------------------
S1 = QQ[x,y]
```

```
J = ideal(x^2,x*y)
M = comodule J
HH^0(M)
(saturate J)/J
prune oo
HH^0(module J)

------------------------------------
-- Local cohomology II -------------
------------------------------------
-- back to the rational quartic
use S
I = ideal"bc-ad,c3-bd2,ac2-b2d,b3-a2c"
N = Hom(I, S^1/I)
N = prune N
res N
HH^0(N)
HH^1(N, Degree=>-5)
HH^2(N, Degree=>-5)
HH^3(N, Degree=>-5)

Ext^1(N,S^{-4})
Ext^2(N,S^{-4})
Ext^3(N,S^{-4})

C = res N
regularity N
betti C

-- Useful function:
Ideal ^ Array := (J,a) -> (
     ell := a#0;
     ideal apply(numgens J, i -> J_i^ell))
J = ideal vars S
J^[2]

-- Another handy function
HF = (M,lo,hi) -> (
     toList apply(lo..hi,
        i -> hilbertFunction(i,M)))
HF(N,-5,5)
```

```
-- Now let's look at approximation
E2 = Ext^2(N,S^{-4})
HF(oo,0,10)

regularity N
  -- so the next line has same
  -- (inverted) HF in degrees >= 0-(1-1)-1 = -1
prune Ext^2(S^1/J,N)
HF(oo,-10,10)
HF(ooo,-10,0)

-- valid for degrees >= -2
prune Ext^2(S^1/J^[2],N)
HF(oo,-10,0)

matrix apply(toList(1..10), i ->
  HF(Ext^2(S^1/J^[i],N),-10,0))

-- maybe powers of the ideal do better?
prune Ext^2(S^1/J^4,N);
HF(oo,-10,0)

-------------------------------------------
-- Omega1 of the Fermat quartic surface --
-------------------------------------------
kk = QQ
S = kk[a..d]
F = ideal(a^4+b^4+c^4+d^4)
R = S/F
p1 = (vars S) ** R
dj = jacobian F ** R
M = homology(p1,dj)
M = prune M
  -- sheaf assocuated to this is
  -- the cotangent sheaf
M = coker lift(presentation M,S)
ann M == F
res M
oo.dd
-- Therefore H^0(Omega1) = 0:
hilbertFunction({0},M)
E2 = Ext^2(M,S^{-4})
```

```
hilbertFunction({0},E2)
E1 = Ext^1(M,S^{-4})
hilbertFunction({0},E1)

X = Proj R
HH^0(OO_X)
HH^1(OO_X)
HH^2(OO_X)

HH^0(sheaf M)
HH^1(sheaf M)
HH^2(sheaf M)

-----------------------------------------
-- Omega2 of the Fermat quartic surface --
-----------------------------------------
M2 = exteriorPower(2,M)
res M2
Omega2 = sheaf M2
HH^1(Omega2)
HH^2(Omega2)
HH^3(Omega2)
HH^0(Omega2(>=0))
use S
J = ideal vars S
ell = 2
E = Hom(J^[ell], M2)

apply(0..10, i -> hilbertFunction(i,M2))
apply(0..10, i -> hilbertFunction(i,E))
apply(0..10, i -> hilbertFunction(i,R))

regularity M2
matrix toList apply(1..12, ell -> (
  HF(Hom(J^[ell], M2), -5, 10)))
prune truncate(0,Hom(J^[7],M2))
prune truncate(-1,Hom(J^[8],M2))
apply(0..10, i -> hilbertFunction(i,R))

---------------------------------------
-- Hodge diamond of a quintic 3-fold --
---------------------------------------
```

```
omega = (I) -> (
     S := ring I;
     R := S/I;
     omegaX := prune homology(vars S ** R,
                       jacobian gens I ** R);
     coker lift(presentation omegaX,S)
     )

load "hodge.m2"

S = QQ[a,b,c,d,e]
I = ideal(a^5+b^5+c^5+d^5+e^5)
omegaX = omega I
res omegaX
cohom (S^1/I)
hodgeDiamond3 I
omega2 = exteriorPower(2,omegaX);
res omega2
betti oo

I = ideal random(S^1, S^{-5})
hodgeDiamond3 I

I = ideal random(S^1, S^{-6})
hodgeDiamond3 I

-- Let's improve the presentation of omega2:
betti omega2
om2sat = coker gens saturate image presentation omega2;
betti om2sat
res om2sat
betti oo

-- now let's improve it even more:
om2B = prune Hom((ideal vars S), om2sat);
betti res om2B

cohom om2B
-- om2B is the best answer.
```